

Generalized Algebraic Data Types in OCaml

Sprachdefinition

$Atom ::= \text{true} \mid \text{false} \mid -^? 0..9^+$

$Expr ::= Atom \mid \text{if } Expr \text{ then } Expr \text{ else } Expr$

Gültige Programme

```
true
```

```
if true then 42 else 0
```

```
if true then  
  if true then 42 else 0  
else  
  0
```

Ungültige Programme

```
if true then 0
```

```
if 0 then true else false
```

```
if true then  
  if true then 42 else 0  
else  
  false
```

In einer Welt ohne GADTs

```
type atom =  
  | Bool of bool  
  | Int  of int  
  
type expr =  
  | Atom of atom  
  | If   of expr * expr * expr
```

Syntaxbaum ohne GADTs

```
let rec eval : expr → atom = function
  | Atom a → a
  | If (c, t, e) → begin match eval c with
      | Bool true → eval t
      | Bool false → eval e
      | _ → failwith "need bool!"
    end
```

Evaluationsfunktion ohne GADTs

```
eval (If
      (Atom (Bool true),
       (Atom (Int 42)),
       (Atom (Int 0))))
```

```
- : atom = Atom (Int 42)
```

```
eval (If
      (Atom (Int 42),
       (Atom (Bool false)),
       (Atom (Int 0))))
```

```
Exception: Failure "need bool!"
```

Beispiele ohne GADTs

- Ungültige Programmdefinition möglich
- Laufzeitfehler im Interpreter
- Zweige können verschiedene Typen haben

In einer Welt mit GADTs

```
type _ atom =  
  | Bool : bool → bool atom  
  | Int  : int  → int  atom  
  
type _ expr =  
  | Atom : 'a atom → 'a expr  
  | If   : bool expr * 'a expr * 'a expr → 'a expr
```

Syntaxbaum mit GADTs

```
let rec eval : .. = function
  | Atom (Bool b) → b
  | Atom (Int i)  → i
  | If (c, t, e)  → if eval c then eval t else eval e
```

Evaluationsfunktion mit GADTs


```
eval (If
      (Atom (Int 42),
       (Atom (Bool true)),
       (Atom (Bool false))))
```

```
eval (If
      (Atom (Bool true),
       (Atom (Bool false)),
       (Atom (Int 42))))
```

Error: Type int is not compatible with type bool

Beispiele mit GADTs

- Keine ungültigen Programmdefinitionen
- Keine Laufzeitfehler im Interpreter
- Exhaustive Patternmatching

Locally Abstract Types

```
let rec eval (type a) (e : a expr) : a = match e with  
| Atom (Bool b) → b  
| Atom (Int i) → i  
| If (c, t, e) → if eval c then eval t else eval e
```

Error: This expression has type a expr but an
expression was expected of type bool expr

Evaluationsfunktion mit GADTs

Polymorphic Recursion

```
let rec eval : 'a. 'a expr → 'a =  
  fun (type a) (e : a expr) : a → match e with  
    | Atom (Bool b) → b  
    | Atom (Int i)   → i  
    | If (c, t, e)   → if eval c then eval t else eval e
```

Evaluationsfunktion mit GADTs

Syntactic Sugar

```
let rec eval : type a. a expr → a = function
| Atom (Bool b) → b
| Atom (Int i)  → i
| If (c, t, e)  → if eval c then eval t else eval e
```

Evaluationsfunktion mit GADTs

Existential Types

```
type stringable =  
  Stringable : {  
    item: 'a;  
    to_string: 'a → string  
  } → stringable
```

```
let print (Stringable s) =  
  print_endline (s.to_string s.item)
```

Unterschiedliche Rückgabewerte

```
type (_, _) mode =  
  | Unsafe : ('a, 'a) mode  
  | Option  : ('a, 'a option) mode
```

```
let first : type a r. a list → (a, r) mode → r =  
  fun lst mde → match lst with  
    | hd :: tl → (match mde with  
      | Unsafe → hd  
      | Option → Some hd)  
    | [ ] → (match mde with  
      | Unsafe → failwith "list is empty"  
      | Option → None)
```

Limitationen

- Typinferenz unentscheidbar
→ Typannotationen notwendig
- `|`-Patterns nicht auflösbar
→ Auflösen und Logik auslagern
- `[@@-deriving ..]` nicht ausdrückbar
→ Non-GADT Version benötigt

In einer Nussschale

- GADTs erlaubt Konstrukturen verschiedene Typenparameter einzusetzen
- Stärkere Aussagen auf Typebene möglich
- Patternmatching nutzt die zusätzlichen Informationen
- Generalisierung von Existentials
- Typinferenz wird unentscheidbar

Folien & Code

github.com/Mari-W/ocaml-gadts

Literatur

- [Real World OCaml: GADTs](#)
Yaron Minsky, Anil Madhavapeddy 2021
- [Detecting use-cases for GADTs in OCaml](#)
Mads Hartmann 2015
- [Stanford CS242: Programming Languages](#)
Will Crichton 2019

Variants vs GADTs

```
type atom =  
  | Bool : bool → atom  
  | Int  : int  → atom
```

atom ohne GADTs

```
type _ atom =  
  | Bool : bool → bool atom  
  | Int  : int  → int  atom
```

atom mit GADTs

Konstrukturen können *verschiedene*
Typenparameter einsetzen.